Pression dans un liquide

Exercice 1

```
(* p0 pression à la surface en mbar *)
(* Δh incertitude sur la profondeur en cm *)
(* ρ masse volumique du liquide en kg/dm³ *)
(* g accélération terrestre en m/s² *)
dn = {p0 -> 780, Δh -> 0.2, ρ -> 1.25, g -> 9.81};
ListPlot[data = Table[{h + (-1)^RandomInteger[{0, 1}] Δh /. dn,
     .1 ρ * g (h + (-1)^RandomInteger[{0, 1}] Δh /. dn) + p0 /. dn}, {h, 0, 20, 2}],
 FrameLabel → {"profondeur h en cm", "pression p en mbar"},
 GridLines → Automatic, Frame → True]
Fit[data, {1, h}, h]
Coefficient[%, h]
10^4 % / g /. dn
```




```
780.057 + 1.21618 h

1.21618

1239.73
```

Exercice 2

Exprimez la pression atmosphérique normale en Pa et en bar. En haute altitude, on mesure une pression de 52 cm Hg. Exprimez cette pression en Pa et en mbar.
La pression atmosphérique normale est la pression moyenne au niveau de la mer. Elle vaut 760 mm Hg.
On obtient l'expression de cette pression p en calculant la force F exercée par une colonne de mercure de cette hauteur sur une surface S :

$$p = \frac{F}{S} = \frac{mg}{S} = \frac{\rho V g}{S} = \rho g h$$

En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient la pression en Pa :

```
dn = {ρ -> 13 590, g → 9.81, h → 0.76};
ρ * g * h /. dn
% / 100
```

```
101 322.
```

```
1013.22
```

Exercice 3

Les chenilles d'un bulldozer reposent sur le sol sur une longueur de 6 m chacune. Quelle doit être leur largeur si l'engin a une masse de 18 tonnes et que la pression sur le sol ne doit pas dépasser 0.49 bar ?
Exprimons la surface totale afin que la pression ne dépasse pas la valeur autorisée :

$$2\,S = \frac{F}{p} = \frac{\mathrm{mg}}{p}$$

La surface d'une chenille est donnée par S = a × b où a est la longueur et b la largeur de la chenille. On obtient donc, pour l'expression de la largeur b :

$$\frac{\mathrm{mg}}{2\,\mathrm{ap}}$$

En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient la largeur en m :

```
dn = {m -> 18 000, g → 9.81, a → 6, p → 49 000};
m * g / (2 a * p) /. dn
```

```
0.300306
```

Exercice 4

Une patineuse a une masse de 50 kg. Les lames de ses patins ont une largeur de 2 mm. Quelle est la pression - en Pa et en bar - exercée sur la glace lorsqu'elle se tient sur un pied, si la lame est en contact avec la glace sur une longueur de 20 cm ?
Exprimons la pression p pour une surface S donnée par S = ab :

$$p = \frac{F}{S} = \frac{\mathrm{mg}}{S} = \frac{\mathrm{mg}}{\mathrm{ab}}$$

En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient la largeur en m :

```
dn = {m → 50, g → 9.81, a → 0.2, b → 0.002};
m * g / (a * b) /. dn
% / 10 ^ 5
```

1.22625×10^{6}

```
12.2625
```

Exercice 5

Au pied d'une colline, un baromètre indique une pression de 75 cm de mercure. Calculez la pression - en Pa et en bar - au sommet de la colline, situé 200 m plus haut. Masse volumique de l'air : 1.293 kg/m3.
Comme la différence d'altitude est faible, nous faisons l'hypothèse que la pression varie linéaire-

ment sur cette distance. Nous pouvons donc exprimer la pression p en fonction de l'altitude h de la manière suivante :

$p_0 - \rho g h$

p_0 est la pression au pied de la colline. Il faut l'exprimer en Pa (voir ex. 2). ρ est la masse volumique de l'air et h est la différence d'altitude. En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient :

```
dn = {p0 → 99 988.4, ρ → 1.293, g → 9.81, h → 200};
p0 − ρ * g * h /. dn
% / 10 ^ 5
```

```
97 451.5
```

```
0.974515
```

Exercice 6

On a construit un tube de Torricelli avec de la gly-cérine. La masse volumique de ce liquide est de 1.26 kg/dm3. Quelle est la hauteur de la colonne de glycérine si la pression atmosphérique vaut 760 mm Hg ? Quel peut être l'intérêt d'un tel baromètre ?
La hauteur h de la colonne de liquide dans un tube de Torri-celli peut s'exprimer à partir de la pression p exercée sur la surface libre du liquide, de la masse volumique ρ du liquide et de l'accélération terrestre g :

$$\frac{p}{\rho g}$$

En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient la hauteur h en m :

```
dn = {p → 101 325, ρ → 1260, g → 9.81};
p / (ρ * g) /. dn
```

```
8.19742
```

L'intérêt d'un tube de Torricelli utilisant un liquide dont la masse volumique est faible est sa sensibilité. Le tube permet alors de mettre en évidence de faibles variations de pression.

Exercice 7

Un bloc de pierre a une masse de 120 kg. Sa base, horizontale, ferme un tuyau dont le diamètre vaut 20 cm. Ce tuyau communique avec un petit tuyau vertical d'un diamètre de 2 cm. Le système contient de l'eau. Initialement, dans chaque branche du tube, l'eau atteint la même hauteur. Quelle quantité faut-il en ajouter dans le petit tuyau pour que le bloc soit soulevé ?
Le bloc commence a être soulevé lorsque la force F = pS due à la pression de l'eau est égale à son poids mg. Cette pression p peut s'exprimer à partir de la hauteur h de la colonne de liquide dans le petit tuyau. Connaissant cette hauteur h et la section s du petit tuyau, on peut trouver le volume V d'eau qu'il faut ajouter.

$$pS = mg$$

$$\rho g h S = mg$$

$$h = \frac{mg}{\rho g S}$$

$$V = sh = \frac{smg}{\rho g S}$$

En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient le volume V en m^3, puis en litre:

```
dn = {m → 120, S → Pi * 0.1^2, s → Pi * .01^2, ρ → 1000, g → 9.81};
s * m * g / (ρ * g * S) /. dn
% * 1000
0.0012

1.2
```

Exercice 8

Dans une maison à six étages, le réseau de distribution d'eau est équipé de robinets dont l'orifice a un diamètre intérieur de 1.2 cm. Au sixième étage, si l'on veut retenir l'eau en appliquant son pouce contre l'ouverture d'un robinet, on doit exercer une force de 30 N. Quelle force doit-on exercer si l'on veut faire cela au rez-de-chaussée ? La hauteur d'un étage est de 3 m.
La pression p en fonction de la profondeur h est donnée par :

$p = p_0 + \rho g h$

p_0 est la pression au sixième étage, ρ est la masse volumique de l'eau et h est la hauteur entre le sixième et le rez-de-chaussée. La force F qu'il faut exercer au rez-de-chaussée s'exprime par :

$F = pS = (p_0 + \rho g h)\, S$

En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient :

```
dn = {S → Pi * 0.006^2, F → 30, ρ → 1000, g → 9.81, h → 6 * 3};
F + ρ * g * h * S /. dn
49.9707
```

Exercice 9

Un cycliste de 50 kg roule sur une bicyclette de 10 kg. Le poids est également réparti sur les roues. Chaque pneu est en contact avec le sol sur une surface de 10 cm2. Calculez la pression de gonflage des pneus.
Chaque roue supporte la moitié du poids. La force F = pS due à la pression de gonflage est égale à la moitié du poids. Elle s'exprime par :

$$F = \frac{mg}{2} = pS$$

$$p = \frac{mg}{2\,S}$$

En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient la pression en Pa :

```
dn = {m → 60, g → 9.81, S → 10^-3};
m * g / (2 S) /. dn
% / 10^5
294 300.

2.943
```

Exercice 10

Un ballon parfaitement souple est gonflé à la pression 1.2 bar (au-dessus de la pression atmosphérique), et est posé sur le sol. Un homme de 80 kg se tient sur le ballon. Calculer le rayon de la

portion de ballon qui est en contact avec le sol.
La force F exercée par la portion d'aire S en contact avec le sol est donnée par F = mg = pS. En supposant que cette portion forme un disque circulaire, mg = πr^2p, on peut exprimer le rayon r de ce disque :

$$r = \sqrt{\frac{\mathrm{mg}}{\pi \mathrm{p}}}$$

En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient le rayon en m :

```
dn = {m → 80, g → 9.81, p → 1.2 * 10^5};
Sqrt[m * g / (Pi * p)] /. dn
% 100
```

```
0.0456262

4.56262
```

Exercice 11

Une plaque de liège flotte sur de l'eau. Elle a une épaisseur de 1 cm et une aire de base de 100 cm2. La masse volumique du liège est de 0.25 g/cm3. Un objet de 60 g repose sur cette plaque de liège, la laissant horizontale. Calculez la hauteur immergée.
Le poids du dispositif est égal à la force d'Archimède :

$$(\rho_1\,\mathrm{Se} + m)\,g = \rho_2\,\mathrm{Shg}$$

$$h = \frac{1}{\rho_2}\left(\rho_1\,e + \frac{m}{S}\right)$$

En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient la hauteur immergée h en m :

```
dn = {ρ1 → 250, ρ2 → 1000, e → 0.01, m → 0.06, S → .01};
1 / ρ2 (ρ1 * e + m / S) /. dn
```

```
0.0085
```

Exercice 12

Une personne de 60 kg se trouve sur un radeau de bois flottant sur l'eau. Les dimensions du radeau sont 3 m, 4 m et 10 cm. La masse volumique du bois est de 0.9 kg/dm3. Calculez la hauteur immergée. De combien la hauteur immergée varie-t-elle lorsque la personne quitte le radeau ?
Le poids du dispositif est égal à la force d'Archimède :

$$(\rho_1\,\mathrm{abe} + m)\,g = \rho_2\,\mathrm{abhg}$$

$$h = \frac{1}{\rho_2}\left(\rho_1\,e + \frac{m}{\mathrm{ab}}\right)$$

En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient la hauteur immergée h en m dans les deux cas ainsi que la variation :

```
dn = {ρ1 → 900, ρ2 → 1000, a → 3, b → 4, e → 0.1, m → 60};
1 / ρ2 (ρ1 * e + m / (a * b)) /. dn
1 / ρ2 (ρ1 * e + m / (a * b)) /. m → 0 /. dn
%% - %
```

```
0.095
```

```
0.09
```

```
0.005
```

Exercice 13

Un ballon est retenu au sol par une corde. Il est gonflé à l'hydrogène (masse volumique : 0,09 kg/m3. L'enveloppe est une sphère de 4 m de rayon. Elle a une masse de 50 kg et soutient une nacelle de 200 kg. Calculer la force avec laquelle est tendue la corde si le ballon se trouve dans de l'air dont la masse volumique vaut 1.293 kg/m3.
La somme des forces exercées sur le ballon doit être nulle.

$\rho_1 \mathrm{Vg} = (m + \rho_2 V) g + T$

La force T exercée par la corde s'exprime par :
En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient la hauteur immergée h en m dans les deux cas ainsi que la variation :

$\mathrm{T} = (\rho_1\text{-}\rho_2)\mathrm{Vg} - \mathrm{mg} = \frac{4}{3}\pi r^3(\rho_1\text{-}\rho_2)\mathrm{g} - \mathrm{mg}$

En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient la force T exercée par la corde en newton :

```
dn = {ρ1 → 1.293, ρ2 → 0.09, g → 9.81, m → 250, r → 4};
4 / 3 Pi * r^3 (ρ1 - ρ2) g - m * g /. dn
```

```
711.258
```

Exercice 14

Un ballon de foire a un diamètre de 30 cm. Le fil qui le retient exerce sur lui une force de 0.08 N. L'enveloppe a une masse de 4.5 g. La masse volumique de l'air est de 1.293 kg/m3. Déduisez de ces données la masse volumique du gaz contenu dans le ballon.
La somme des forces exercées sur le ballon doit être nulle.

$\rho_1 \mathrm{Vg} = \mathrm{mg} + \rho_2 \mathrm{Vg} + T$

La masse volumique ρ_2 s'exprime par :

$$\rho_2 = \frac{\mathrm{V}\rho_1 g - \mathrm{mg} - T}{\mathrm{Vg}} = \frac{\left(\frac{4}{3}\pi r^3 \rho_1 - m\right) g - T}{\frac{4}{3}\pi r^3 g}$$

En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient la masse volumique ρ_2 du gaz contenu dans le ballon en kg/m^3 :

```
dn = {ρ1 → 1.293, r → 0.15, g → 9.81, T → 0.08, m → 0.0045};
(4 / 3 Pi * r^3 ρ1 - m - T / g) / (4 / 3 Pi * r^3) /. dn
```

```
0.397846
```

Exercice 15

On veut construire une balise sphérique apte à flotter sur l'eau. Son rayon est fixé à 20 cm, et elle

ne doit s'enfoncer dans l'eau que de 6 cm. Quelle masse faut-il lui donner ?
Le volume obtenu par révolution d'une courbe y = f(x) autour de l'axe des abscisses est donné par :

$$V = \pi \int_{x_{min}}^{x_{max}} y^2 \, dx$$

Pour une demi-sphère de rayon r, la fonction y = f(x) est celle décrivant un quart de cercle centré sur l'origine dont l'équation est $x^2 + y^2 = r^2$. Nous pouvons donc écrire :

$$V = \pi \int_0^r (r^2 - x^2) \, dx = \frac{2}{3} \pi r^3$$

Pour obtenir le volume de la calotte sphérique immergée, il suffit de soustraire au volume de la demi-sphère le volume obtenu en intégrant de 0 à r-h, où h est la profondeur d'immersion :

$$V_{calotte} = \frac{2}{3} \pi r^3 - \pi \int_0^{r-h} (r^2 - x^2) \, dx$$

En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient le volume de la calotte sphérique en m^3. Ce volume permet de calculer la force d'Archimède, qui est égale au poids de la balise en N, ce qui permet de trouver sa masse en kg :

```
dn = {r → 0.2, h → 0.06, ρ → 1000, g → 9.81};
2 / 3 Pi * r ^ 3 - Pi * Integrate[r ^ 2 - x ^ 2, {x, 0, r - h}]
% /. dn
% * ρ * g /. dn
% / g /. dn
```

$$\frac{2 \pi r^3}{3} - \pi \left(\frac{h^3}{3} - h^2 r + \frac{2 r^3}{3}\right)$$

```
0.00203575

19.9707

2.03575
```

Exercice 16

Une boule de bois flotte sur l'eau. Elle émerge d'une hauteur égale aux trois dixièmes de son rayon. Quelle est la masse volumique de ce bois ?
Si la boule flotte, la somme des forces qui agissent sur elle est nulle. Son poids est compensé par la force d'Archimède. Nous pouvons donc écrire :

$$P = F_A$$

$$\rho_1 V g = \rho_2 V_{immergé} \, g$$

Le volume immergé est égal au volume de la sphère moins le volume de la calotte émergée :

$$\rho_1 V = \rho_2 (V - V_{calotte})$$

Le volume de la calotte sphérique émergée s'obtient en soustrayant au volume de la sphère, le volume obtenu en intégrant l'expression de -r à r - h, où h est la hauteur d'émersion :

$$V_{calotte} = \frac{4}{3} \pi r^3 - \pi \int_{-r}^{r-h} (r^2 - x^2) \, dx$$

La masse volumique de la matière dont la boule est faite s'exprime donc par :

$$\rho_1 = \left(1 - \frac{V_{calotte}}{V}\right) \rho_2$$

En remplaçant les grandeurs par leurs valeurs numériques exprimées dans les unités du SI, on obtient la masse volumique de la matière dont la boule est faite en kg/m^3 :

```
dn = {h → 3 / 10, ρ2 → 1000., r → 1};
(1 - (4 / 3 Pi * r ^ 3 - Pi * Integrate[r ^ 2 - x ^ 2, {x, -r, r - h}]) / (4 / 3 Pi * r ^ 3)) ρ2 /. dn
```

939.25

```
dn = {h → 3 / 10, ρ2 → 1000, r → 1.};
(2 Pi / 3 * r ^ 3 + Pi * Integrate[r ^ 2 - x ^ 2, {x, 0, r - h}]) ρ2 / (4 Pi * r ^ 3 / 3) /. dn
```

939.25

```
sol = Solve[
  ρ1 * 4 * Pi * r ^ 3 / 3 == ρ2 (2 * Pi * r ^ 3 / 3 + Pi * Integrate[r ^ 2 - x ^ 2, {x, 0, r - h}]), ρ2]
```

$$\left\{\left\{\rho 2 \to \frac{4\, r^3\, \rho 1}{(h - 2\, r)^2\, (h + r)}\right\}\right\}$$

```
dn = {r → 1, ρ1 → 1, h → 3 / 10}
sol[[1, 1, 2]] /. dn
% // N
1 / %
```

$$\left\{r \to 1,\ \rho 1 \to 1,\ h \to \frac{3}{10}\right\}$$

$$\frac{4000}{3757}$$

1.06468

0.93925

Sphère flottante

En faisant tourner la fonction $f(x)$ autour de l'axe x, on délimite le volume d'une sphère de rayon r = 1 donné par :

```
Pi * Integrate[r ^ 2 - (x - r) ^ 2 /. r → 1, {x, 0, 2 r - h}]
% /. {r → 1, h → 0}
```

$$\frac{1}{3} \pi\, (h - 2\, r)^2\, (3 + h - 2\, r)$$

$$\frac{4\, \pi}{3}$$

```
Plot[Sqrt[r^2 - (x - r)^2] /. r → 1, {x, 0, 2},
 AspectRatio → Automatic, AxesLabel → {"x", "f(x)"}]
```



Reportons le volume immergé en fonction de la hauteur émergée *h* lorsque cette sphère flotte sur un liquide :

$$\text{Plot}\left[\frac{1}{3}\pi\,(h-2\,r)^2\,(3+h-2\,r)\;/.\;r\to 1,\;\{h,\,0,\,2\},\;\text{AxesLabel}\to\{\text{"}h\text{"},\,\text{"Volume immergé"}\}\right]$$



Si la sphère flotte, son poids est égal à la poussée d'Archimède. Exprimons le rapport *q* de la masse volumique de la sphère ρ_1 à celle du liquide ρ_2 en fonction de la hauteur émergée *h* :

```
sol = Solve[q == Integrate[r^2 - (x - r)^2, {x, 0, 2 r - h}] / (4 * r^3 / 3), q]
sol[[1, 1, 2]] /. {r → 1, h → .3} (* vérification sol pb 16 *)
Plot[sol[[1, 1, 2]] /. r → 1, {h, 0, 2},
 GridLines → {Range[0, 2, .1], Range[0, 1, .1]}, AxesLabel → {"h", "ρ1/ρ2"}]
```

$$\left\{\left\{q \to \frac{h^3 - 3 h^2 r + 4 r^3}{4 r^3}\right\}\right\}$$

0.93925



Fonction réciproque

```
sol = Solve[q == (h^3 - 3 h^2 r + 4 r^3)/(4 r^3), h];
sol[[3, 1, 2]] /. r → 1
Plot[sol[[3, 1, 2]] /. r → 1, {q, 0, 1}, AxesLabel → {"ρ1/ρ2", "h"}]
```

$$1 + \left(1 - i \sqrt{3}\right) \Big/ \left(2 \left(1 - 2 q + 2 \sqrt{-q + q^2}\right)^{1/3}\right) + \frac{1}{2} \left(1 + i \sqrt{3}\right) \left(1 - 2 q + 2 \sqrt{-q + q^2}\right)^{1/3}$$



La position *c* du centre de la sphère est donnée par *c* = *h* - *r* :

```
Plot[sol[[3, 1, 2]] - r /. r → 1, {q, 0, 1}, AxesLabel → {"ρ₁/ρ₂", "c"}]
```




```
Manipulate[
 Graphics3D[{Sphere[{0, 0, n = 1/2 (2 + (1 - ⅈ √3) / (1 - 2 q + 2 √((-1 + q) q))^(1/3) + (1 + ⅈ √3)
        (1 - 2 q + 2 √((-1 + q) q))^(1/3)) - 1}, 1], Opacity[0.5],
   Polygon[{{-2, -2, 0}, {2, -2, 0}, {2, 2, 0}, {-2, 2, 0}}]}, PlotLabel →
   Style["hauteur émergée " <> ToString[NumberForm[Chop[n + 1], {3, 3}]] <> " R",
    "Label"], Boxed → False],
 {{q, .5, "rapport ρ₁/ρ₂"}, 0, 1, .001, Appearance → "Labeled"}]
```